Chapter 10

Dates and Times in DS2

DS2 Dates, Times, and Timestamps

Overview of DS2 Dates, Times, and Timestamps

DS2 supports the SQL style date and time conventions that are used in other data sources. When your data source is a not a SAS data set, DS2 can process dates and times that have a data type of DATE, TIME, and TIMESTAMP.

Date and time values with a data type of DATE, TIME, and TIMESTAMP can be converted to a SAS date, time, or datetime value. SAS date, time, or datetime values can be converted to a value having a DATE, TIME, or TIMESTAMP data type.

DS2 provides date and time functions that convert any date or time value to SAS date, time, and datetime values, and back again to a recognizable date or time value. For more information, see “Date, Time, and Datetime Functions” on page 77.

The date and time intervals that are supported in ANSI SQL are not supported in DS2.

Declaring Date, Time, and Timestamp Variables

You declare a date, time, or timestamp variable by using the DATE, TIME, or TIMESTAMP data types in the DECLARE statement, as in this example:

```
dcl date dt;
dcl time tm;
dcl timestamp tmstmp;
```

Note: If you use a precision when you declare a time or timestamp variable, the time or timestamp values are not rounded to the specified precision until they are generated by the DATA statement. Internally, the time or timestamp constant values are simply copied to the time or timestamp variable.

Note: If you are working with TIME and TIMESTAMP values in a data source other than SAS and you do not specify a precision, the default precision is the DS2 default precision of 0 for TIME and 6 for TIMESTAMP.

For additional information about the DS2 date and time data types, see Chapter 5, "DS2 Data Types," on page 33.

DS2 Date, Time, and Timestamp Values

Once you declare a date, time, or timestamp variable, the value of the variable can be only a DS2 date, time, or timestamp constant that has the following syntax:

DATE'*yyyy-mm-dd*'

TIME'*hh*:*nn*:*ss*[.*fraction*]'

TIMESTAMP'*yyyy-mm-dd hh*:*nn*:*ss*[.*fraction*]'

where

- *yyyy* is a four-digit year
- *mm* is a two-digit month, 01–12
- *dd* is a two-digit day, 01–31
- *hh* is a two-digit military hour, 00–23
- *nn* is a two-digit minute, 00–59
- *ss* is a two-digit second, 00–60
- *fraction* can be one to nine digits, 0–9, is optional, and represent a fraction of a second

The string portion of the value after the DATE, TIME, or TIMESTAMP keyword must be enclosed in single quotation marks.

In the date constant, the hyphens are required and the length of the date string must be 10.

In the time constant, the colons are required. If the fraction of a second is not present, the time string must be 8 characters long and exclude the period. DS2 issues an error if the period is present without a fraction. If the fraction of second is present, the fraction can be up to 9 digits long and the time string can be up to 18 characters long (including the period).

In the timestamp constant, the hyphens in the date are required as well as the colons in the time. If the fraction of a second is not present, the timestamp string must be 19 characters long and exclude the period. If the fraction of a second is present, the fraction can be up to 9 digits long and the timestamp string can be up to 29 characters long.

Here are some examples of DS2 date, time, and timestamp constants:

```
date'2012-01-31'
time'20:44:59'
timestamp'2012-02-07 07:00:00.7569'
```

Operations on DS2 Dates and Times

The only operations that can be performed on DATE, TIME, and TIMESTAMP values are operations that use the relational operators <, >, <=, >=, =, ^=, and IN, such as in the following statement:

```
if tm in(time'10:22:31', time'12:55:01') then
   if tm < time'13:30:00' then put 'Early afternoon';
   else put 'Time not available';
```

DS2 does not calculate date and time intervals on values that have the data types of DATE, TIME, and TIMESTAMP.

SAS Date, Time, and Datetime Values

A SAS date value is the number of days between January 1, 1960 and a specified date. Dates before January 1, 1960 are negative numbers; dates after are positive numbers. For example, the SAS date value for January 1, 1960 is 0, -365 for January 1, 1959, and 17532 for January 1, 2008.

A SAS time value is the number of seconds since midnight of the current day. SAS time values are between 0 and 86400.

A SAS datetime value is the number of seconds between January 1, 1960 and a specific hour, minute, and second of a specific date.

When a numeric column is read from a SAS data set and the numeric column has a SAS date, time, or datetime format associated with it, the column is converted to a DS2 type DATE, TIME, or TIMESTAMP. If the numeric column in a SAS data set does not have a format or has a format that is not a SAS date, time, or datetime format, the column is processed as type DOUBLE.

All calculations on dates and times are done as a SAS date value, a SAS time value, or a SAS datetime value. For more information, see "Date, Time, and Datetime Functions" on page 77.

After calculations are complete, there are other functions that can then format the SAS date, time, and datetime values to recognizable date and time formats.

Converting SAS Date, Time, and Datetime Values to a DS2 Date, Time, or Timestamp Value

SAS date, time, and datetime values can be converted to DS2 dates, time, and timestamp values by using the TO_DATE, TO_TIME, and TO_TIMESTAMP functions. The argument of these functions is any value or expression that represents a SAS date, time, or datetime value and has a type DOUBLE. You can then use either the PUT statement or a format in the DECLARE statement to format the date, time, or timestamp value.

Here is an example.

```
data _null_;
   dcl date ds2d having format YYMMDD10.;
   dcl time ds2t having format TIME18.9;
```

```
   dcl timestamp ds2dt having format DATETIME28.9;
   dcl double d t ts;
   method init();
      d = 19358;
      ds2d = to_date(d);
      ds2t= to_time(d);
      ds2dt= to_timestamp(d);
      put ds2d ds2t ds2dt;
   end;
enddata;
run;
```

The following lines are written to the SAS log.

```
2012-12-31
 5:22:38.000000000
01JAN1960:05:22:38.000000000
```

For more information, see the "TO_DATE Function" in *SAS Viya: DS2 Language Reference*, the "TO_TIME Function" in *SAS Viya: DS2 Language Reference*, and the "TO_TIMESTAMP Function" in *SAS Viya: DS2 Language Reference*.

Converting DS2 Date, Time, and Timestamp Values to SAS Date, Time, or Datetime Values

DS2 date, time, and timestamp values can be converted to a SAS datetime value by using the TO_DOUBLE function. This function converts the date, time, or timestamp CHAR or NCHAR string to a SAS datetime value with a data type of DOUBLE. You can then use any DS2 format to display the value in a date, time, or datetime format.

The following DS2 program illustrates how you can convert a DS2 timestamp to a SAS date, time, and datetime values:

```
data _null_;
    method run();
       dcl timestamp DS2ts;
       dcl double sasdtval sasd sastm;
       dcl char(28) fmtdate fmttime fmtdt;
       DS2ts = timestamp '2012-06-04 10:54:34.012';
       put DS2ts;
       sasdtval = to_double(DS2ts);
       sasd = datepart(sasdtval);
       sastm = timepart(sasdtval);
       put sasdtval:best16.7;
       put sasd:best.;
       put sastm:best.;
       fmtdate = put(sasd, yymmdd10.);
       fmttime = put(sastm, time.);
       fmtdt = put(sasdtval, datetime21.7);
       put fmtdate=;
       put fmttime=;
         put fmtdt=;
    end;
enddata;
```